Computational Companion to “Generalized Quasi-Symmetric Nets: A Constructive Approach to the Equimodular Elliptic Type of n × m Kokotsakis Polyhedra” — Helper

A. Nurmatov, M. Skopenkov, F. Rist, J. Klein, D. L. Michels
Tested on: Mathematica 14.0

Section 1

Results of Section 1 is used in Lemma 1 (see paper)

In[1197]:=

```
(* ======================================================================*)
(*PART 1:INITIALIZATION& DOMAIN ASSUMPTIONS*)
(* ======================================================================*)
ClearAll[α, β, γ, δ, σ, ᾱ, β̄, γ̄, δ̄, a, b, c, d, r, s, f, M, assumptions, isTrueQ];

(*Core angular relationships*)
σ = (α + β + γ + δ) / 2;

ᾱ = σ - α;
β̄ = σ - β;
γ̄ = σ - γ;
δ̄ = σ - δ;

(*Sign tracking*)
ε = Abs[Sin[σ]] / Sin[σ];

(*Geometric domain constraints*)
```

```
assumptions = {0 < α < Pi, 0 < β < Pi, 0 < γ < Pi,
   0 < δ < Pi, 0 < ᾱ < Pi, 0 < β̄ < Pi, 0 < γ̄ < Pi, 0 < δ̄ < Pi};

(* ============================================================================*)
(*PART 2:SYSTEM VARIABLES*)
(* ============================================================================*)

a = Sin[α] / Sin[ᾱ];
b = Sin[β] / Sin[β̄];
c = Sin[γ] / Sin[γ̄];
d = Sin[δ] / Sin[δ̄];

r = a * d;
s = c * d;
f = a * c;
M = a * b * c * d;

(* ============================================================================*)
(*PART 3:FORMULA DEFINITIONS*)
(* ============================================================================*)

(*---Computational Definitions---*)
rhsOneMinusM =
  (Sin[σ] * Sin[ᾱ - β] * Sin[γ̄ - β] * Sin[δ̄ - β]) / (Sin[ᾱ] * Sin[β̄] * Sin[γ̄] * Sin[δ̄]);

rhsRMinusOne = (Sin[σ] * Sin[γ̄ - β]) / (Sin[ᾱ] * Sin[δ̄]);

rhsSMinusOne = (Sin[σ] * Sin[ᾱ - β]) / (Sin[γ̄] * Sin[δ̄]);

rhsOneMinusMOverf = (Sin[σ] * Sin[δ̄ - β]) / (Sin[β̄] * Sin[δ̄]);

rhsMainFormula =
  ε * Sqrt[rhsOneMinusMOverf * rhsRMinusOne * rhsSMinusOne / rhsOneMinusM] *
    Sin[δ̄] /. Sqrt[Csc[1 / 2 (-α - β - γ - δ) + δ] ^2 * Sin[1 / 2 (α + β + γ + δ)] ^2] →
    Abs[Sin[1 / 2 (α + β + γ + δ)]] * Csc[δ̄];


(*---Visual Definitions---*)
visOneMinusM = HoldForm[
   (Sin[σ] * Sin[ᾱ - β] * Sin[γ̄ - β] * Sin[δ̄ - β]) / (Sin[ᾱ] * Sin[β̄] * Sin[γ̄] * Sin[δ̄])];

visRMinusOne = HoldForm[(Sin[σ] * Sin[γ̄ - β]) / (Sin[ᾱ] * Sin[δ̄])];

visSMinusOne = HoldForm[(Sin[σ] * Sin[ᾱ - β]) / (Sin[γ̄] * Sin[δ̄])];
```

```
visOneMinusMOverf = HoldForm[(Sin[σ] * Sin[δ̄ - β]) / (Sin[β̄] * Sin[δ̄])];

visMainFormula =
  HoldForm[ε * Sqrt[((f - M) * (r - 1) * (s - 1)) / (f * (1 - M))] * Sin[δ̄]];

(* ======================================================================*)
(*PART 4:VALIDATION& REPORTING*)
(* ======================================================================*)

(*Testing function evaluating under global assumptions*)
isTrueQ[lhs_, rhs_] :=
  TrueQ[FullSimplify[lhs == rhs, Assumptions → assumptions]];

(*Define the validation suite*)
testSuite =
  {{"Intermediate Formula 1", HoldForm[1 - M], 1 - M, visOneMinusM, rhsOneMinusM},
   {"Intermediate Formula 2", HoldForm[r - 1], r - 1, visRMinusOne, rhsRMinusOne},
   {"Intermediate Formula 3", HoldForm[s - 1], s - 1, visSMinusOne, rhsSMinusOne},
   {"Intermediate Formula 4", HoldForm[(f - M) / f], (f - M) / f,
    visOneMinusMOverf, rhsOneMinusMOverf}, {"Main Formula Verification",
    HoldForm[Sin[σ]], Sin[σ], visMainFormula, rhsMainFormula}};

(*Process results into table rows*)
tableRows = Map[{#[[1]], Item[TraditionalForm[#[[2]]], Alignment → Center],
     Item[TraditionalForm[#[[4]]], Alignment → Center],
     If[isTrueQ[#[[3]], #[[5]]], Style["✓ Valid", Darker[Green], Bold],
      Style["× Failed", Red, Bold]]} &, testSuite];

(*Generate the Grid output*)
Grid[Prepend[tableRows, {"Description",
   "Left Hand Side (LHS)", "Right Hand Side (RHS)", "Equality Status"}],
 Frame → All, FrameStyle → Directive[Gray, Thickness[1]],
 Alignment → {{Left, Center, Center, Center}, Center},
 Spacings → {2, 2.5}, Background → {None, {1 → LightGray}}, ItemStyle →
  {Directive[FontFamily → "Helvetica", 12], {1 → Directive[Bold, 14]}}]
```

Out[1226]=

Description	Left Hand Side (LHS)	Right Hand Side (RHS)	Equality Status
Intermediate Formula 1	$1 - M$	$\frac{\sin(\sigma)\sin(\bar{\alpha}-\beta)\sin(\bar{\gamma}-\beta)\sin(\bar{\delta}-\beta)}{\sin(\bar{\alpha})\sin(\bar{\beta})\sin(\bar{\gamma})\sin(\bar{\delta})}$	✓ Valid
Intermediate Formula 2	$r - 1$	$\frac{\sin(\sigma)\sin(\bar{\gamma}-\beta)}{\sin(\bar{\alpha})\sin(\bar{\delta})}$	✓ Valid
Intermediate Formula 3	$s - 1$	$\frac{\sin(\sigma)\sin(\bar{\alpha}-\beta)}{\sin(\bar{\gamma})\sin(\bar{\delta})}$	✓ Valid
Intermediate Formula 4	$\frac{f-M}{f}$	$\frac{\sin(\sigma)\sin(\bar{\delta}-\beta)}{\sin(\bar{\beta})\sin(\bar{\delta})}$	✓ Valid
Main Formula Verification	$\sin(\sigma)$	$\varepsilon\sqrt{\frac{(f-M)(r-1)(s-1)}{f(1-M)}}\sin(\bar{\delta})$	✓ Valid